/*
Joanna Pepin	
Proposal title: Public Perception of Money in Families

HYPOTHESES

Stated-Hyp1: "It is likely that most people perceive married couples as exemplifying family unity, treating income as a collective resourse, while there may be less of an expectation of family unity for cohabiting couples....Parental status also seems to
matter for the likelihood of sharing financial resources, with parents more alike than non-parents, no matter their marital status." (p.2)

	Test-Hyp1: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: bank accounts]
	
	Test-Hyp2: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: bank accounts]	


	Test-Hyp3: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: fairness perception]
	
	Test-Hyp4: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: fairness perception]	


	Test-Hyp5: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: unexpected income treatment]
	
	Test-Hyp6: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: unexpected income treatment]	


	Test-Hyp7: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: personal spending]
	
	Test-Hyp8: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: personal spending]	
	
	
Stated-Hyp2: "hypothesize higher-earners will be evaluated as having more rights to personal spending money (money which can be spent without consultation) and greater say in decision making" (p. 3)

	Test-Hyp9: Expectation of complete resource sharing will be lower if the person who wants to spend the money is the higher earner vs. lower earner. [personal spending scenario]

	Test-Hyp10: Expectation of complete resource sharing will be lower if the person who makes the final decision is the higher earner vs. lower earner. [decision making scenario]	

Stated-Hyp3: "women's breadwinning may be insufficient to elicit notions of breadwinning, as evidenced by a lack of agreement that higher-earning women are entitled to more personal spending money or increased decision making power within the family. Perceptions of gendered breadwinning may also be conditional on marital status..." (p. 3-4)

	Test-Hyp11: Expectation of complete resource sharing will be lower if the person who wants to spend the money is the higher earner vs. lower earner, but this effect be moderated by gender [negative interaction of high earner and woman]. [personal spending scenario]

	Test-Hyp12: Expectation of complete resource sharing will be lower if the person who makes the final decision is the higher earner vs. lower earner, but this effect be moderated by gender [negative interaction of high earner and woman]. [decision making scenario]	
	
	
********************************************************************************
NOTES:	
- outcomes are mentioned on page 5: 
"[1] how money should be arranged, [2] their perception of fairness in hypothetical decision making scenarios, [3] how unexpected income should be treated, and [4] to report opinions of individual rights to personal spending money". It is unclear how the author exactly planned to measure these outcomes. 
	
- For outcome 1, the measure of how to arrange money in shared/separate bank 
accounts is categorical. 

- For outcome 2, there are 2 fairness perception questions. The first question 
already anchored responses, so we are using that.  

- For outcome 4, the most relevant measure is used.   
*/

clear all
use "pepin791.dta", clear


* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* which gender earns more (MODERATOR)
	tab dov_earnings
	recode dov_earnings (1=2 "Anthony earns more") (2=1 "Michelle earns more") (3=0 "earn equally"), gen(treat_earnmore)
	tab treat_earnmore

* woman earns more
	recode treat_earnmore (1=1) (2=0) (*=.), gen(treat_womanearnmore)
	tab treat_womanearnmore
	
* marital status 
	recode dov_relstat (1=1) (2=0), gen(treat_married)
	tab treat_married

* parent status  
	recode dov_parentst (1=0) (2=1), gen(treat_parent)
	tab treat_parent
	
* duration of relationship (3 or 7 years) 
	recode dov_reldur (1=0) (2=1), gen(treat_longduration)
	tab treat_longduration
	// ghost treatment
	
	
* high-earner makes decision (spending scenario 1)
	tab dov_person_b04
	gen highearner_decides=.
	replace highearner_decides=1 if dov_person_b04==1 & treat_earnmore==1
	replace highearner_decides= 1 if dov_person_b04==2 & treat_earnmore==2
	replace highearner_decides=0 if dov_person_b04==1 & treat_earnmore==2
	replace highearner_decides= 0 if dov_person_b04==2 & treat_earnmore==1
	replace highearner_decides=0 if treat_earnmore==0
	tab highearner_decides, mis
		
* high-earner wants personal spending money	
	tab dov_person_b10
	gen highearner_personalspend=.
	replace highearner_personalspend=1 if dov_person_b04==1 & treat_earnmore==1
	replace highearner_personalspend= 1 if dov_person_b04==2 & treat_earnmore==2
	replace highearner_personalspend=0 if dov_person_b04==1 & treat_earnmore==2
	replace highearner_personalspend= 0 if dov_person_b04==2 & treat_earnmore==1
	replace highearner_personalspend=0 if treat_earnmore==0
	tab highearner_personalspend, mis	
	
* OUTCOME MEASURES

* 1. how money should be arranged: bank account ownership

	/*
	Couples organize their income in many different ways. Do you think Michelle and Anthony should: 

	1. Have a shared account in which they both deposit all their earned income
	2. Keep all their earned income in separate, individual accounts
	3. Have both a shared account and separate, individual accounts
	
	NOTE: author didn't specify how to split the categories. I'm using a binary split for complete vs. incomplete resource sharing
	*/

	tab b01 
	recode b01 (1=1 "complete sharing") (2/3=0 "incomplete sharing")  (-1=.), gen(completeshare_bankac)
	tab completeshare_bankac
	

*2. perception of fairness in hypothetical decision making scenario 

	/*
	B04. [S]
	Anthony and Michelle have to make a decision about [INSERT DOV_Item TEXT]. Although they talked this over and they care equally about the decision, [INSERT DOV_Person_B04 TEXT] ultimately decides since they can't agree. How fair do you think it is that [INSERT DOV_Person_B04 TEXT] made the final decision?

	- note: ignoring b06. see notes above.
	*/

	foreach var in b04 {
		replace `var'=. if `var'==-1
		tab `var', mis
	}
	// higher values indicate higher expectation that decisions should be made collectively
	
	gen completeshare_fairdecisions=b04
	
*3. how unexpected income should be treated  
	/*
	B08. [S] 
	If [INSERT DOV_Person_B08 TEXT] receives an unexpected bonus at work, who should get to decide what to do with the additional income?
	*/

	tab b08
	recode b08 (-1=.) (1/2=0) (3=1), gen(completeshare_bonus)
	tab completeshare_bonus, mis

*4. personal spending money
	/*
	B10. [S]
	[INSERT DOV_Person_B10 TEXT]'s cell phone is functioning but [IF DOV_Person_B10=1, INSERT `she'; IF DOV_Person_B10=2, INSERT `he'] wants a new phone. Because [IF DOV_Person_B10=1, INSERT `she'; IF DOV_Person_B10=2, INSERT `he'] is still in a cell phone contract, upgrading to a new phone is expensive and cannot be replaced for a lower cost for another six months. Who should decide whether [IF DOV_Person_B10=1, INSERT `she'; IF DOV_Person_B10=2, INSERT `he'] gets a new phone now?
	*/	
	
	tab b10
	recode b10 (-1=.) (1/2=0) (3=1), gen(completeshare_personalspending)
	tab completeshare_personalspending, mis
	
********************************************************************************
*ANALYSIS	

*Test-Hyp1: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: bank accounts]
	
	reg completeshare_bankac i.treat_married
	// do not reject. 0.000
	tess 1.treat_married +, init(Pepin791) bonf(4)
	
*Test-Hyp2: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: bank accounts]	

	reg completeshare_bankac i.treat_parent
	// do not reject. 0.000
	tess 1.treat_parent +, bonf(4)	

*Test-Hyp3: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: fairness perception]
	
	reg completeshare_fairdecisions i.treat_married
	// reject. 0.156	
	tess 1.treat_married +, bonf(4)
	
*Test-Hyp4: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: fairness perception]	

	reg completeshare_fairdecisions i.treat_parent
	// reject. 0.576	
	tess 1.treat_parent +, bonf(4)	

*Test-Hyp5: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: unexpected income treatment]
	reg completeshare_bonus i.treat_married
	// do not reject. 0.000	
	tess 1.treat_married +, bonf(4)
	
*Test-Hyp6: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: unexpected income treatment]	
	reg completeshare_bonus i.treat_parent
	// do not reject. 0.001
	tess 1.treat_parent +, bonf(4)		

*Test-Hyp7: Expectation of complete resource-sharing will be higher for married vs. unmarried couples. [outcome: personal spending]
	reg completeshare_personalspending i.treat_married
	// do not reject. 0.000		
	tess 1.treat_married +, bonf(4)
	
*Test-Hyp8: Expectation of complete resource-sharing will be higher for parent vs. non-parent couples. [outcome: personal spending]	
	reg completeshare_personalspending i.treat_parent
	// do not reject. 0.000		
	tess 1.treat_parent +, bonf(4)		

	
*Test-Hyp9: Expectation of complete resource sharing will be lower if the person who wants to spend the money is the higher earner vs. lower earner. [personal spending scenario]
	reg completeshare_fairdecisions i.highearner_decides if treat_earnmore!=0
	// reject. 0.616
	tess 1.highearner_decides -
	
*Test-Hyp10: Expectation of complete resource sharing will be lower if the person who makes the final decision is the higher earner vs. lower earner. [decision making scenario]	
	reg completeshare_personalspending i.highearner_personalspend if treat_earnmore!=0
	// reject. 0.863
	tess 1.highearner_personalspend -
	
*Test-Hyp11: Expectation of complete resource sharing will be lower if the person who wants to spend the money is the higher earner vs. lower earner, but this effect be moderated by gender [negative interaction of high earner and woman]. [personal spending scenario]
	reg completeshare_fairdecisions i.highearner_decides##i.treat_womanearnmore 
	// do not reject. 0.000
	tess 1.highearner_decides#1.treat_womanearnmore  -

*Test-Hyp12: Expectation of complete resource sharing will be lower if the person who makes the final decision is the higher earner vs. lower earner, but this effect be moderated by gender [negative interaction of high earner and woman]. [decision making scenario]	
	reg completeshare_personalspending i.highearner_personalspend##i.treat_womanearnmore 
	// reject. 0.626
	tess 1.highearner_personalspend#1.treat_womanearnmore  -	
		